
4 Local Descent

Up to this point, we have focused on optimization involving a single design
variable. This chapter introduces a general approach to optimization involving
multivariate functions, or functions with more than one variable. The focus of this
chapter is on how to use local models to incrementally improve a design point
until some convergence criterion is met. We begin by discussing methods that,
at each iteration, choose a descent direction based on a local model and then
choose a step size. We then discuss methods that restrict the step to be within a
region where the local model is believed to be valid. This chapter concludes with
a discussion of convergence conditions. The next two chapters will discuss how
to use first- and second-order models built from gradient or Hessian information.

4.1 Descent Direction Iteration

A common approach to optimization is to incrementally improve a design point
x by taking a step that minimizes the objective value based on a local model. The
local model may be obtained, for example, from a first- or second-order Taylor
approximation. Optimization algorithms that follow this general approach are
referred to as descent direction methods. They start with a design point x(1) and
then generate a sequence of points, sometimes called iterates, to converge to a
local minimum.1 1 The choice of x(1) can affect the

success of the algorithm in find-
ing a minimum. Domain knowl-
edge is often used to choose a rea-
sonable value. When that is not
available, we can search over the
design space using the techniques
that will be covered in chapter 13.

The iterative descent direction procedure involves the following steps:

1. Check whether x(k) satisfies the termination conditions. If it does, terminate;
otherwise proceed to the next step.

2. Determine the descent direction d(k) using local information such as the gradient
or Hessian. Some algorithms assume ‖d(k)‖ = 1, but others do not.

3. Determine the step size or learning rate α(k). Some algorithms attempt to
optimize the step size so that the step maximally decreases f .2 2 We use step size to refer to the

magnitude of the overall step. Ob-
taining a new iterate using equa-
tion (4.1) with a step size α(k) im-
plies that the descent direction d(k)

has unit length. We use learning
rate to refer to a scalar multiple
used on a descent direction vec-
torwhich does not necessarily have
unit length.

4. Compute the next design point according to:

x(k+1) ← x(k) + α(k)d(k) (4.1)

There are many different optimization methods, each with their own ways of
determining α and d.

4.2 Line Search

For the moment, assume we have chosen a descent direction d, perhaps using one
of the methods discussed in one of the subsequent chapters. We need to choose
the step factor α to obtain our next design point. One approach is to use line search,
which selects the step factor that minimizes the one-dimensional function:

minimize
α

f (x + αd) (4.2)

Line search is a univariate optimization problem, which was covered in chapter 3.
We can apply the univariate optimization method of our choice.3 To inform the

3 The Brent-Dekker method, men-
tioned in the previous chapter, is
a commonly used univariate op-
timization method. It combines
the robustness of the bisection
method with the speed of the se-
cant method.

search, we can use the derivative of the line search objective, which is simply the
directional derivative along d at x+ αd. Line search is demonstrated in example 4.1
and implemented in algorithm 4.1.

function line_search(f, x, d)
objective = α -> f(x + α*d)
a, b = bracket_minimum(objective)
α = minimize(objective, a, b)
return x + α*d

end

Algorithm 4.1. A method for con-
ducting a line search, which finds
the optimal step factor along a
descent direction d from design
point x to minimize function f.
The minimize function can be im-
plemented using a univariate op-
timization algorithm such as the
Brent-Dekker method.

One disadvantage of conducting a line search at each step is the computational
cost of optimizing α to a high degree of precision. Instead, it is common to quickly
find a reasonable value and then move on, selecting x(k+1), and then picking a
new direction d(k+1).

Some algorithms use a fixed step factor. Large steps will tend to result in faster
convergence but risk overshooting the minimum. Smaller steps tend to be more
stable but can result in slower convergence. A fixed step factor α is sometimes
referred to as a learning rate.

Another method is to use a decaying step factor:

α(k) = α(1)γk−1 for γ ∈ (0, 1] (4.3)

Decaying step factors are especially popular when minimizing noisy objective
functions,4 and are commonly used in machine learning applications.

4 We will discuss optimization in
the presence of noise and other
forms of uncertainty in chapter 17.

Consider conducting a line search on f (x1, x2, x3) = sin(x1x2) + exp(x2 +

x3)− x3 from x = [1, 2, 3] in the direction d = [0,−1,−1]. The corresponding
optimization problem is:

minimize
α

sin((1 + 0α)(2− α)) + exp((2− α) + (3− α))− (3− α)

which simplifies to:

minimize
α

sin(2− α) + exp(5− 2α) + α− 3

The minimum is at α ≈ 3.127 with x ≈ [1,−1.126,−0.126].

Example 4.1. Line search used to
minimize a function along a de-
scent direction.

4.3 Approximate Line Search

It is often more computationally efficient to perform more iterations of a descent
method than to do exact line search at each iteration, especially if the function
and derivative calculations are expensive. Many of the methods discussed so far
can benefit from using approximate line search to find a suitable step size with a
small number of evaluations. Since descent methods must descend, a step size α

may be suitable if it causes a decrease in the objective function value. However, a
variety of other conditions may be enforced to encourage faster convergence.

The condition for sufficient decrease5 requires that the step size cause a sufficient 5 This condition is sometimes re-
ferred to as the Armijo condition.decrease in the objective function value:

f (x(k+1)) ≤ f (x(k)) + βα∇d(k) f (x(k)) (4.4)

with β ∈ [0, 1] often set to β = 1× 10−4. Figure 4.1 illustrates this condition. If
β = 0, then any decrease is acceptable. If β = 1, then the decrease has to be at
least as much as what would be predicted by a first-order approximation.

0

f (x) + βα∇d f (x)
f (x) + α∇d f (x)

f (x + αd)

sufficient decrease

α

y

Figure 4.1. The sufficient decrease
condition, the firstWolfe condition,
can always be satisfied by a suffi-
ciently small step size along a de-
scent direction.

If d is a valid descent direction, then there must exist a sufficiently small step
size that satisfies the sufficient decrease condition. We can thus start with a large
step size and decrease it by a constant reduction factor until the sufficient decrease
condition is satisfied. This algorithm is known as backtracking line search6 because

6 Also known as Armijo line search,
L. Armijo, “Minimization of Func-
tions Having Lipschitz Continu-
ous First Partial Derivatives,” Pa-
cific Journal of Mathematics, vol. 16,
no. 1, pp. 1–3, 1966.

of how it backtracks along the descent direction. Backtracking line search is shown
in figure 4.2 and implemented in algorithm 4.2. We walk through the procedure
in example 4.2.

function backtracking_line_search(f, ∇f, x, d, α; p=0.5, β=1e-4)
y, g = f(x), ∇f(x)
while f(x + α*d) > y + β*α*(g⋅d)

α *= p
end
return α

end

Algorithm 4.2. The backtracking
line search algorithm, which takes
objective function f, its gradient
∇f, the current design point x, a
descent direction d, and the maxi-
mum step size α. We can optionally
specify the reduction factor p and
the first Wolfe condition parameter
β.

Note that the cdot character ⋅
aliases to the dot function such that
a⋅b is equivalent to dot(a,b). The
symbol can be created by typing
\cdot and hitting tab.

The first condition is insufficient to guarantee convergence to a local minimum.
Very small step sizes will satisfy the first condition but can prematurely converge.
Backtracking line search avoids premature convergence by accepting the largest
satisfactory step size obtained by sequential downscaling and is guaranteed to
converge to a local minimum.

Another condition, called the curvature condition, requires the directional deriva-
tive at the next iterate to be shallower:

∇d(k) f (x(k+1)) ≥ σ∇d(k) f (x(k)) (4.5)

Figure 4.2. Backtracking line
search used on the Rosenbrock
function (appendix B.6). The black
lines show the seven iterations
taken by the descent method and
the red lines show the points con-
sidered during each line search.

where σ controls how shallow the next directional derivative must be. Figures 4.3
and 4.4 illustrate this condition. It is common to set β < σ < 1 with σ = 0.1 when
approximate linear search is used with the conjugate gradient method and to 0.9

when used with Newton’s method.7 7 The conjugate gradient method is
introduced in section 5.2, andNew-
ton’s method is introduced in sec-
tion 6.1.

An alternative to the curvature condition is the strong curvature condition, which
is amore restrictive criterion in that the slope is also required not to be too positive:

|∇d(k) f (x(k+1))| ≤ −σ∇d(k) f (x(k)) (4.6)

Figure 4.5 illustrates this condition.
Together, the sufficient decrease condition and the curvature condition form

the Wolfe conditions. The sufficient decrease condition is often called the first Wolfe
condition and the curvature condition is called the second Wolfe condition. The
sufficient decrease condition with the strong curvature condition form the strong
Wolfe conditions.

Satisfying the strong Wolfe conditions requires a more complicated algorithm,
strong backtracking line search (algorithm 4.3).8 The method operates in two phases. 8 J. Nocedal and S. J.Wright,Numer-

ical Optimization, 2nd ed. Springer,
2006.The first phase, the bracketing phase, tests successively larger step sizes to bracket

an interval [α(k−1), α(k)] guaranteed to contain step lengths satisfying the Wolfe
conditions.

Consider approximate line search on f (x1, x2) = x2
1 + x1x2 + x2

2 from
x = [1, 2] in the direction d = [−1,−1], using a maximum step size of 10,
a reduction factor of 0.5, a first Wolfe condition parameter β = 1× 10−4, and
a second Wolfe condition parameter σ = 0.9.

We checkwhether themaximumstep size satisfies the firstWolfe condition,
where the gradient at x is g = [4, 5]:

f (x + αd) ≤ f (x) + βα(g⊤d)

f ([1, 2] + 10 · [−1,−1]) ≤ 7 + 1× 10−4 · 10 · [4, 5]⊤[−1,−1]

217 ≤ 6.991

It is not satisfied.
The step size is multiplied by 0.5 to obtain 5, and the first Wolfe condition

is checked again:

f ([1, 2] + 5 · [−1,−1]) ≤ 7 + 1× 10−4 · 5 · [4, 5]⊤[−1,−1]

37 ≤ 6.996

It is not satisfied.
The step size is multiplied by 0.5 to obtain 2.5, and the firstWolfe condition

is checked again:

f ([1, 2] + 2.5 · [−1,−1]) ≤ 7 + 1× 10−4 · 2.5 · [4, 5]⊤[−1,−1]

3.25 ≤ 6.998

The first Wolfe condition is satisfied.
The candidate design point x′ = x + αd = [−1.5,−0.5] is checked against

the second Wolfe condition:

∇d f (x′) ≥ σ∇d f (x)

[−3.5,−2.5]⊤[−1,−1] ≥ σ [4, 5]⊤[−1,−1]

6 ≥ −8.1

The second Wolfe condition is satisfied.
Approximate line search terminates with x = [−1.5,−0.5].

Example 4.2. An example of back-
tracking line search, an approxi-
mate line search method.

0

minimum of second-order approximation

α

y

Figure 4.3. The curvature con-
dition, the second Wolfe con-
dition, is necessary to ensure
that second-order function approx-
imations have positive curvature,
thereby having a unique global
minimum.

0

f (x) + α∇
d f (x)

f (x) + ασ∇
d f (x)

reduced curvature

α

y

Figure 4.4. Regions where the cur-
vature condition is satisfied.

0

f (x) + α∇
d f (x)

f (x) + ασ∇
d f (x)

α

y

Figure 4.5. Regions where the
strong curvature condition is sat-
isfied.

An interval guaranteed to contain step lengths satisfying the Wolfe conditions
is found when one of the following conditions hold:

f (x + αd) ≥ f (x) (4.7)
f (x + αd) > f (x) + βα∇d f (x) (4.8)
∇ f (x + αd) ≥ 0 (4.9)

0

f (x) + βα∇
d f (x)

y = f (x)

α

y

f (x + αd) ≥ f (x)

f (x + αd) > f (x) + βα∇d f (x)

∇ f (x + αd) ≥ 0

Wolfe conditions satisfied

Figure 4.6. The interval [0, α] is
guaranteed to bracket an interval
containing a step length satisfying
the strong Wolfe conditions when
any of these three conditions is
true.

Satisfying equation (4.8) is equivalent to violating the first Wolfe condition,
thereby ensuring that shrinking the step length will guarantee a satisfactory step
length. Similarly, equation (4.7) and equation (4.9) guarantee that the descent
step has overshot a localminimum, and the region betweenmust therefore contain
a satisfactory step length.

Figure 4.6 shows where each bracketing condition is true for an example line
search. The figure shows bracket intervals [0, α], whereas advanced backtracking
line search successively increases the step length to obtain a bracketing interval
[α(k−1), α(k)].

In the zoom phase, we shrink the interval to find a step size satisfying the
strong Wolfe conditions. The shrinking can be done using the bisection method
(section 3.7), updating the interval boundaries according to the same interval
conditions. This process is shown in figure 4.7.

α(1)

α(2)

α(3)

α(4)

α(5)

α

y

f (x + αd) ≥ f (x)

f (x + αd) > f (x) + βα∇d f (x)

∇ f (x + αd) ≥ 0

Figure 4.7. The first phase of strong
backtracking line search, indicated
with black open circles, is used to
bracket an interval. In this case, the
condition ∇ f (x + α(4)d) ≥ 0 is
triggered, causing the bracketing
interval to be [α(3), α(4)]. Then the
zoom phase, indicated by the red
open circle, shrinks the bracketed
region until a suitable step length
is found.4.4 Trust Region Methods

Descent methods can place too much trust in their first- or second-order infor-
mation, which can result in excessively large steps or premature convergence. A
trust region9 is the local area of the design space where the local model is believed 9 K. Levenberg, “A Method for

the Solution of Certain Non-Linear
Problems in Least Squares,” Quar-
terly of Applied Mathematics, vol. 2,
no. 2, pp. 164–168, 1944.

to be reliable. A trust region method, or restricted step method, maintains a local
model of the trust region that both limits the step taken by traditional line search
and predicts the improvement associated with taking the step. If the improve-
ment closely matches the predicted value, the trust region is expanded. If the
improvement deviates from the predicted value, the trust region is contracted.10 10 A recent review of trust re-

gion methods is provided by Y.X.
Yuan, “Recent Advances in Trust
Region Algorithms,” Mathemati-
cal Programming, vol. 151, no. 1,
pp. 249–281, 2015.

Figure 4.8 shows a design point centered within a circular trust region.
Trust region methods first choose the maximum step size and then the step

direction, which is in contrast with line search methods that first choose a step
direction and then optimize the step size. A trust region approach finds the
next step by minimizing a model of the objective function f̂ over a trust region
centered on the current design point x. An example of f̂ is a second-order Taylor
approximation (see appendix C.2). The radius of the trust region, δ, is expanded
and contracted based on how well the model predicts function evaluations. The
next design point x′ is obtained by solving:

minimize
x′

f̂ (x′)

subject to
∥
∥x− x′

∥
∥ ≤ δ

(4.10)

where the trust region is defined by the positive radius δ and a vector norm.11

11 There are a variety of efficient
methods for solving equa-
tion (4.10) efficiently. For an
overview of the trust region
method applied to quadratic
models, see D.C. Sorensen,
“Newton’s Method with a Model
Trust Region Modification,” SIAM
Journal on Numerical Analysis,
vol. 19, no. 2, pp. 409–426, 1982.

The equation above is a constrained optimization problem, which is covered in
chapter 10.

function strong_backtracking(f, ∇, x, d; α=1, β=1e-4, σ=0.1)
y0, g0, y_prev, α_prev = f(x), ∇(x)⋅d, NaN, 0
αlo, αhi = NaN, NaN

# bracket phase
while true

y = f(x + α*d)
if y > y0 + β*α*g0 || (!isnan(y_prev) && y ≥ y_prev)

αlo, αhi = α_prev, α
break

end
g = ∇(x + α*d)⋅d
if abs(g) ≤ -σ*g0

return α
elseif g ≥ 0

αlo, αhi = α, α_prev
break

end
y_prev, α_prev, α = y, α, 2α

end

# zoom phase
ylo = f(x + αlo*d)
while true

α = (αlo + αhi)/2
y = f(x + α*d)
if y > y0 + β*α*g0 || y ≥ ylo

αhi = α
else

g = ∇(x + α*d)⋅d
if abs(g) ≤ -σ*g0

return α
elseif g*(αhi - αlo) ≥ 0

αhi = αlo
end
αlo = α

end
end

end

Algorithm 4.3. Strong backtrack-
ing approximate line search for
satisfying the strong Wolfe condi-
tions. It takes as input the objective
function f, the gradient function
∇, the design point x and direction
d from which line search is con-
ducted, an initial step size α, and
the Wolfe condition parameters
β and σ. The algorithm’s bracket
phase first brackets an interval con-
taining a step size that satisfies the
strong Wolfe conditions. It then re-
duces this bracketed interval in the
zoom phase until a suitable step
size is found. We interpolate with
bisection, but other schemes can be
used.

Figure 4.8. Trust region methods
constrain the next step to lie within
a local region. The trusted region
is expanded and contracted based
on the predictive performance of
models of the objective function.

The trust region radius δ is expanded or contracted based on the local model’s
predictive performance. Trust region methods compare the predicted improve-
ment ∆ypred = f (x)− f̂ (x′) to the actual improvement ∆yact = f (x)− f (x′):

η =
actual improvement

predicted improvement =
f (x)− f (x′)
f (x)− f̂ (x′)

(4.11)

The ratio η is close to 1 when the predicted step size matches the actual step
size. If the ratio is too small, such as below a threshold η1, then the improvement
is considered sufficiently less than expected, and the trust region radius is scaled
down by a factor γ1 < 1. If the ratio is sufficiently large, such as above a threshold
η2, then our prediction is considered accurate, and the trust region radius is
scaled up by a factor γ2 > 1. Algorithm 4.4 provides an implementation and
figure 4.9 visualizes the optimization procedure. Example 4.3 shows how to
construct noncircular trust regions.

4.5 Termination Conditions

There are four common termination conditions for descent direction methods:

• Maximum iterations. We may want to terminate when the number of iterations
k exceeds some threshold kmax. Alternatively, we might want to terminate once
a maximum amount of elapsed time is exceeded.

k > kmax (4.12)

function trust_region_descent(f, ∇f, H, x, k_max;
η1=0.25, η2=0.5, γ1=0.5, γ2=2.0, δ=1.0)
y = f(x)
for k in 1 : k_max

x′, y′ = solve_trust_region_subproblem(∇f, H, x, δ)
η = (y - f(x′)) / (y - y′) # improvement ratio
if η < η1

δ *= γ1 # scale down trust region
else

x, y = x′, y′ # accept new point
if η > η2

δ *= γ2 # scale up trust region
end

end
end
return x

end

using Convex, SCS
function solve_trust_region_subproblem(∇f, H, x0, δ)

x = Variable(length(x0))
p = minimize(∇f(x0)⋅(x-x0) + quadform(x-x0, H(x0))/2)
p.constraints += norm(x-x0) <= δ
solve!(p, SCS.Optimizer)
return (x.value, p.optval)

end

Algorithm 4.4. The trust region
descent method, where f is the
objective function, ∇f produces
the derivative, H produces the
Hessian, x is an initial design
point, and k_max is the number of
iterations. The optional parameters
η1 and η2 determine when the
trust region radius δ is increased
or decreased, and γ1 and γ2
control the magnitude of the
change. An implementation for
solve_trust_region_subproblem
must be provided that solves
equation (4.10). We have provided
an example implementation
that uses a second-order Taylor
approximation about x0 with a
circular trust region.

Figure 4.9. Trust region optimiza-
tion used on the Rosenbrock func-
tion (appendix B.6).

Trust regions need not be circular. In some cases, certain directions may have
higher trust than others.

A norm can be constructed to produce elliptical regions:

‖x− x0‖E = (x− x0)
⊤E(x− x0)

with ‖x− x0‖E ≤ 1 where E is a symmetric matrix that defines the ellipse.

The ellipse matrix E can be updated with each descent iteration, which
can involve more complicated adjustments than scaling the trusted region.

Example 4.3. Trust region op-
timization need not use circular
trust regions. Additional detail is
provided by J. Nocedal and S. J.
Wright, “Trust-Region Methods,”
inNumerical Optimization. Springer,
2006, pp. 66–100.

• Absolute improvement. This termination condition looks at the change in the
function value over subsequent steps. If the change is smaller than a given
threshold, it will terminate:

f (x(k))− f (x(k+1)) < ǫa (4.13)

• Relative improvement. This termination condition also looks at the change in
function value but uses the step factor relative to the current function value:

f (x(k))− f (x(k+1)) < ǫr| f (x(k))| (4.14)

• Gradient magnitude. We can also terminate based on the magnitude of the
gradient:

‖∇ f (x(k+1))‖ < ǫg (4.15)

In cases where multiple local minima are likely to exist, it can be beneficial to
incorporate random restarts after our termination conditions are met where we
restart our local descent method from randomly selected initial points.

4.6 Summary

• Descent direction methods incrementally descend toward a local optimum.

• Univariate optimization can be applied during line search.

• Approximate line search can be used to identify appropriate descent step sizes.

• Trust regionmethods constrain the step to lie within a local region that expands
or contracts based on predictive accuracy.

• Termination conditions for descent methods can be based on criteria such as
the change in the objective function value or magnitude of the gradient.

4.7 Exercises

Exercise 4.1. Why is it important to have more than one termination condition?

Exercise 4.2. The first Wolfe condition requires

f (x(k) + αd(k)) ≤ f (x(k)) + βα∇d(k) f (x(k)) (4.16)

What is the maximum step length α that satisfies this condition, given that f (x) =

5 + x2
1 + x2

2, x(k) = [−1,−1], d = [1, 0], and β = 10−4?

